



SB_3.1: Modell-Parameter eines MOS-FET




Bild SB_3.1_1: Schaltung zum Simulationsbeispiel 3.1

*** Netzliste SB_3.1 ***

```
V_UDS      D  0  10V
V_UGS      G  0  3V
M_V1       D  G  0  0  IRF150-X1
.MODEL     IRF150-X1  NMOS (VTO=2    L=2u   W=3m)
.DC        LIN V_UDS  0  15  1m
+ V_UGS    LIST  3  4  5  6
.TEMP      20
.PROBE
.END
```

Abschn.: **3.5**

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: ID (M_V1)

OK



SB_3.2: Kleinsignalverstärker mit MOS-FET




Bild SB_3.2_1: Schaltung zum Simulationsbeispiel 3.2

*** Netzliste SB_3.2 ***

```
V_UB      2      0      15V
V_U1      ein    0      AC    15mV SIN 0 15m 1kHz 0 0 0
M_V1      3      1      4      4      IRF150-X
R_R10     1      2      180k
R_R20     1      0      120k
R_RD0     2      3      43
R_RS0     4      0      7.5
R_Ra      aus    0      100k
C_C1      ein    1      47n
C_C2      3      aus    10n
C_C3      0      4      {Cvar}
.MODEL    IRF150-X  NMOS (LEVEL=3  VTO=2  L=2u  W=3m  CBD=3.2n)
.PARAM    Cvar=20u
.AC       DEC 1000  10  100MEG
.STEP    PARAM  Cvar LIST 10u  50u  100u  200u
.PROBE
.END
```

Abschn.: 3.5

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (aus)

OK

Ende dieses Beispiels



SB_3.3: Analogschalter mit S-FET




Bild SB_3.3_1: Schaltung zum Simulationsbeispiel 3.3

*** Netzliste SB_3.3 ***

```
V_U1      3      0      AC      1V
+SIN      0      200mV    100Hz 0 0 0
V_UP      St      0
+PULSE    0      -2      50m    1p    1p    150m 300m
J_V1      2      1      0      J2N4393
R_RV      3      2      100k
R_RSt     St      1      47k
C_CSt     0      1      1u
R_Ra      0      2      10k
.TRAN     0      0.5    0      10u
.TEMP     27
.PROBE
.END
```

Abschn.: **3.5**

- DC-Analyse
- AC-Analyse
- Tran.-Analyse**
- DC-Main-Sweep
- DC-Nested-Sweep
- AC-Sweep
- Param.-Sweep
- Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: **V (2)**

OK



SB_3.4: MOS-FET als Schalter




Bild SB_3.4_1: Schaltung zum Simulationsbeispiel 3.4

*** Netzliste SB_3.4 ***

```
V_UB      1      0      15V
V_U1     ein     0      10V
M_V1     aus     ein    0      0      IRF155-X
R_RD0    aus     1      {Rvar}
R_Ra     0      aus    1k
.MODEL   IRF155-X  NMOS (LEVEL=3  VTO=2  L=2u
+ W=3m    CBD=3.2n)
.PARAM   Rvar=2k
.DC      LIN  V_U1  0      6      1m
+ PARAM  Rvar  LIST  500  1k    2k    5k
.TEMP    27
.PROBE
.END
```

Abschn.: **3.5**

DC-Analyse

AC-Analyse

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (2)

OK



SB_3.5: CMOS-Inverter




Bild SB_3.5_1: Schaltung zum Simulationsbeispiel 3.5

*** Netzliste SB_3.5 ***

```
V_UB      1      0      5V
V_UP      ein    0
+PULSE    0V    5V    500n  0      0      2.5u  5u
M_V1      aus    ein    0      0      IRF150
M_V2      aus    ein    1      1      IRF9140
R_Ra      0      aus    100k
.TRAN     0      15u    0      1n
.TEMP     27
.PROBE
.END
```

Abschn.: **3.5**

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: **V (aus)**

OK



SB_3.6: Modell-Parameter eines bipolaren Transistors (BC 550 C)




Bild SB_3.6_1: Schaltung zum Simulationsbeispiel 3.6 (aus Lehrbeispiel 3.5)

*** Netzliste SB_3.6 ***

```
V_UCE      C      0      1V
I_IB       0      B      DC    10uA
Q_V1       C      B      0      BC550C
.DC        LIN    V_UCE    0V    20V    1mV
+ LIN     I_IB    0      200uA    40uA
.TEMP      27
.PROBE
.END
```

Abschn.: **3.5**

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: IC (Q_V1)

OK



SB_3.7: Stromverstärkung eines bipolaren Transistors (BC 550 C)




Bild SB_3.7_1: Schaltung zum Simulationsbeispiel 3.7

Abschn.: 3.5

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

*** Netzliste SB_3.7 ***

I_IB 0 B DC 10uA

V_UCE C 0 10V

Q_V1 C B 0 BC550C-X1

Q_V2 C B 0 BC550C-X2

Q_V3 C B 0 BC550C-X3

.MODEL BC550C-X1 NPN(IS=7.049f VAF=23.89 BF=493.2 ISE=99.2f NE=1.829 IKF=.1542
+NK=.6339 XTB=1.5 BR=2.886 ISC=7.371p NC=1.508 IKR=5.426 RC=1.175)

.MODEL BC550C-X2 NPN(IS=7.049f VAF=23.89 BF=493.2 ISE=496f NE=1.829 IKF=.1542
+NK=.6339 XTB=1.5 BR=2.886 ISC=7.371p NC=1.508 IKR=5.426 RC=1.175)

.MODEL BC550C-X3 NPN(IS=7.049f VAF=23.89 BF=493.2 ISE=19.84f NE=1.829 IKF=.1542
+NK=.6339 XTB=1.5 BR=2.886 ISC=7.371p NC=1.508 IKR=5.426 RC=1.175)

.DC LIN I_IB 1nA 150uA 10nA

.TEMP 20

.PROBE

.END

Lösung siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: IC (Q_Vx) / IB (Q_Vx)

OK



SB_3.8: Temperaturabhängigkeit von Modell-Parametern (BC 550 C)




Bild SB_3.8_1: Schaltung zum Simulationsbeispiel 3.8

*** Netzliste SB_3.8 ***

```
V_UCE      C 0 10V
I_IB       0 B DC 10uA
Q_V1      C B 0 BC550C
.DC        LIN TEMP      0      100  .1
.STEP      NPN BC550C    (XTB)     LIST 0.5  1.5  2.5
.PROBE
.END
```

Abschn.: **3.5**

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: IC (Q_V1)

OK



SB_3.9: Kleinsignalverstärker mit BJT (BC 550 C)




Bild SB_3.9_1: Schaltung zum Simulationsbeispiel 3.9

*** Netzliste SB_3.9 ***

```
V_UB      4      0      20V
V_Ue     ein    5      AC    10mV
+SIN      0     10mV 10kHz 0 0 0
Q_V1     2      1      3     BC550C
R_RiG    5      0      10
R_R1     1      4     20k
R_R2     0      1     3.6k
R_RC     2      4     240
R_RE     0      3     51
C_Ce     ein    1     6.8u
C_CRE    0      3     470u
C_Ca     2      aus   100n
R_Ra     0      aus   1MEG
.AC      DEC   1000  100  100MEG
.TEMP    27
.PROBE
.END
```

Abschn.: 3.5

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (aus)

OK

Ende dieses Beispiels



SB_3.10: Spannungsstabilisierung mit Längstristor (BD 139)




Bild SB_3.10_1: Schaltung zum Simulationsbeispiel 3.10

*** Netzliste SB_3.10 ***

```
V_Ue      1      0      12V
Q_V1      1      2      3      BD139
D_V2      0      2      D1N750
R_RV      2      1      160
R_Ra      0      3      {Rvar}
.MODEL    D1N750-X
+ D (Is=880.5E-18 Rs=.25 Bv=6.8 Ibv=20.245m Nbv=1.7)
.PARAM    Rvar=100
.DC       LIN    V_Ue 10    14    1m
+ PARAM   Rvar  LIST  700  70    7
.TEMP     27
.PROBE
.END
```

Abschn.: 3.5

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: V (3)

OK



SB_3.11: Differenzverstärker (BC 550 C)




Bild SB_3.11_1: Schaltung zum Simulationsbeispiel 3.11

*** Netzliste SB_3.11 ***

```
V_UB1      V+      0      15V
V_UB2      0      V-     15V
I_I0       2      V-     DC    20mA
V_Ue1     3      0      2V
V_Ue2     1      0      2V
Q_V1      a1     3      2     BC550C
Q_V2      a2     1      2     BC550C
R_RC1     V+     a1     300
R_RC2     V+     a2     300
.DC       LIN   V_Ue1    0      5      1m
.TEMP     27
.PROBE
.END
```

Abschn.: **3.5**

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → Add Trace

Trace-Expression-Zeile: V (a1) V (a2)

OK



SB_3.12: Thyristor als Schalter




Bild SB_3.12_1: Gleichstromschalter im Simulationsbeispiel 3.12

*** Netzliste (1) SB_3.12 ***

```
V_UB      B      0      48V
I_IG      0      1
+ PULSE   0      2mA   100ms   0      0      200ms   10s
X_V1      A      1      0      2N1595
X_Sw1     0      RC     Sw_perClose
+ PARAMS: t_D=1s      t_on=5gs   t_off=5gs   t_switch=1us
+ R_low=1m R_high=1g
C_CY      RC     A      100u   IC=0
R_RY      RC     B      1k
R_RA      B      A      5
.TRAN     0ns   2s     0      1m
.TEMP     27
.PROBE
.END
```

Abschn.: **3.5**

DC-Analyse
AC-Analyse
Tran.-Analyse

DC-Main-Sweep
DC-Nested-Sweep
AC-Sweep
Param.-Sweep
Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: **V (B) – V (A)**

OK




Bild SB_3.12_2: Wechselstromschalter im Simulationsbeispiel 3.12

*** Netzliste (2) SB_3.12 ***

```

V_UB      B      0      SIN      0      45V      50Hz      0      0      0
I_IG      0      1
+PULSE    0      2mA    2ms     0      0      1ms     20ms
X_V1      A      1      0      2N1595
R_RA      B      A      50
.TRAN     0ns    60m    0      10u
.TEMP     27
.PROBE
.END
    
```

Abschn.: [3.5](#)

- DC-Analyse
- AC-Analyse
- Tran.-Analyse**
- DC-Main-Sweep
- DC-Nested-Sweep
- AC-Sweep
- Param.-Sweep
- Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: V (B) – V (A)

OK




Bild SB_3.12_3: Schaltung zur Phasenanschnittsteuerung eines Triac im Simulationsbeispiel 3.12

*** Netzliste (3) SB_3.12 ***

```

V_UB      B      0      SIN      0      120V  50Hz  0  0  0
I_IG1     0      1
+ PULSE   0      50mA      6ms  0      0      1ms  20ms
I_IG2     0      1
+ PULSE   0      -50mA     16ms 0      0      1ms  20ms
X_V1     AK      1      0      2N5444
R_RA      B      AK      5
.TRAN     0ns    60m    0      10u
.TEMP     27
.PROBE
.END
    
```

Abschn.: [3.5](#)

DC-Analyse

AC-Analyse

Tran.-Analyse

DC-Main-Sweep

DC-Nested-Sweep

AC-Sweep

Param.-Sweep

Temp.-Sweep

Lösung siehe: PROBE-Fenster

Trace → *Add Trace*

Trace-Expression-Zeile: V (B) - V (AK)

OK